On the Calculation of HLW Loading in Borosilicate Glass
(File: HLW_glass_description_21Dec2010.doc)

I. Introduction

The calculation of how much high-level-waste can be put into borosilicate glass has many parts. The purpose of this technical note is to tell how the calculation is done and where all the input information comes from. Section II provides an overview of the calculation with an example of calculated results which are generated by a simple BASIC code. Section III describes how elemental isotope composition information from a burned assembly is converted to an oxide composition. The reason for doing this is because elements report to the glass as oxides, not elements (except for a few noble metals), and reactor-performance calculations usually report elemental information. Section IV is a detailed illustration of the calculation of the HLW loading in glass. This section derives an important piece of information, and that is what the mass of process chemicals added during reprocessing is relative to the fission product mass. HLW is not just fission products but contains these other added process chemicals. Then Section V is a listing of the BASIC code that takes into account all of the information described in the following sections and calculates results.

It is worthwhile to note that no single source of information has been found that gives all the information required to calculation the HLW loading that can be put into glass. The major unknown turned out to be the mass of process chemicals added during reprocessing that “ride along” with the fission products and report to vitrification. Requests for a vitrification flow sheet with a material balance and final glass composition were answered with “proprietary.”

The nomenclature changes from time to time in the following text, so be careful when reading about fission products and high-level waste. Fission products are always fission products, but sometimes in the discussion HLW will be fission products plus process chemicals or HLW will be just fission products. Minor actinides and Zr fines also end up in HLW, and minor actinides are taken into account here but Zr fines are not because quantification of these fines is not readily available. HLW takes on different definitions in the open literature also, so careful consideration must always be given to the discussion.

II. HLW Loadings in Borosilicate Glass

The HLW loading from a 60 GWd/ton burnup fuel is calculated according to the information and description that follows. The fission product mass from a spent fuel assembly is used to calculate the fission product oxide mass based on the reported elemental masses in an assembly. From information reported in the open literature an estimate of the process chemicals carried along with the fission products can be obtained to yield the total mass of fission products plus process chemicals which is defined here as the HLW mass to be vitrified. Note that HLW is not just fission products but also the added process chemicals that are needed to perform the separations.

The individual element-to-oxide mass multipliers are documented in Section III: *Fission Product Oxides in Glass.* The element-to-oxide mass multipliers are then used to calculate the fission product mass from specific assembly compositions at whatever burnup the assembly experienced. Two burnups are considered; 40 GWd/ton and 60 GWd/ton in files CSNF_Composition_40GWd.xls, August 16, 2010, and CSNF_Composition_60GWd.xls, August 16, 2010. These spread-sheet files yield an average element-to-mass multiplier, 1.236 and 1.237 respectively. There are upper limit concentrations for MoO_3 and the noble metals (RuO_2 + Pd + Rh + Te) in glass so the spread sheets also calculate the weight % of these in the fission-product oxide mass. This information is then used to calculate the HLW mass that can be loaded into glass and the final glass mass in a specified volume. The glass-loading calculation is described in Section III: *Calculation of HLW Loading in Borosilicate Glass*, August 17, 2010, file: Calculation_HLW_in_Bglass.doc. The calculation is performed by the code GLASSVOL.BAS and is listed in Section V. An example of the input variables to perform this calculation is as follows:

```
Input variables for ID: 60 GWd/ton at 4.5% U235.

1.  Glass density, gm/cc =                                 2.75
2.  HLW oxide loading in glass, wt % =                     9.69

3.  MoO3 loading in HLW oxide mass, wt % =                10.00
4.  Maximum MoO3 loading in final glass, wt % =            1.50

5.  Noble metal loading in HLW oxide mass, wt % =         12.90
6.  Maximum noble metal loading in final glass, wt % =     1.25

7.  Glass canister volume, m^3 =                           0.89
8.  Element-to-oxide mass multiplier =                     1.236
9.  Assembly mass, kilograms =                             430
10. Fission product oxide mass per assembly, kg =          24.9
```

The calculated results for the above input are:

```
Calculated results:

Glass mass in one canister, kilograms =               2447.5
HLW oxide mass in this glass mass, kilograms =         237.1
HLW element mass in this glass mass, kilograms =       191.9, NUWASTE input.
HLW oxide loading in glass used in this calc, wt% =      9.7

MoO3 mass in the HLW oxide mass, kilograms =            23.7
MoO3 concentration in the glass, wt % =                 0.97

Noble metal mass in the HLW oxide mass, kilograms =     30.6
Noble metal concentration in the glass, wt % =          1.25

# of assemblies that yield the fission product mass       10, in 1 canister.
Metric tons initial HM metal in these assemblies =       4.1
Glass volume per metric ton initial HM, liters =       217.3

HLW oxide mass adjusted to accommodate the noble metal glass concentration.
The adjustment factor is 0.707 of the initial input HLW oxide mass.
The initial HLW oxide loading in glass, wt %, was   13.7
```

The calculated HLW elemental mass loading per canister is 0.192 MT and denoted as the NUWASTE input parameter above. This loading is limited by the noble metal loading of 1.25 wt% in the glass as RuO_2 + Pd + Rh + Te. If this noble-metal loading limit is increased to 1.50 wt% the HLW elemental loading per canister increases to 0.230 MT, and if the limit is increased to 2.0 wt% the HLW elemental loading per canister increases to 0.271 MT. This limit of 2.0 wt% is high enough so that the noble metals "fit" in the glass with no adjustment required. These results are summarized here:

Controlling parameter for HLW loading in glass	Value of controlling parameter	HLW elemental mass loading in one canister, kilograms
Noble-metal loading	1.25 wt% maximum in glass	192
Noble-metal loading	1.50 wt% maximum in glass	230
Noble-metal loading	2.00 wt% maximum in glass (not invoked)	271

It is to be noted that the glass composition obtained (and used) from the open literature, F. Frizon, S. Gin, C. Jegou, *Mass Transfer Phenomena in Nuclear Waste Packages*, pages 39-46, in L.Q. Wang (Ed.): *Advances in Transport Phenomena*, 2009, Springer-Verlag, does not discuss concentration limits in glass for MoO_3 and the noble metals (RuO_2 + Pd + Rh + Te). If this composition is such that the HLW plus process chemicals can be accommodated (dissolved) in the glass, then the HLW elemental mass loading in one canister is 0.271 MT.

These glass loadings per canister are canister specific, the 0.89 m^3 canister volume is based on a 2-ft diameter by 10-ft long canister (internal dimensions). If the canister length is increased to 15 feet then the canister volume is 1.34 m^3.

III. Fission Product Oxides in Glass

Fission products are usually reported to a vitrification process as oxides. G. Roth and S. Weisenburger in *Vitrification of high-level liquid waste: glass chemistry, process chemistry and process technology*, Nuclear Engineering and Design, 202 (2000), pp 197-207, state on page 199 that "Usual waste oxide loads are between 15 and 25 wt. % for borosilicate glass." Figure 1 on page 198 presents a waste composition for a high-level waste concentration (HLWC) that is 56% weight % fission products/actinides, 14% corrosion elements, and 30% process chemicals. Horst Wiese and Maurits Demonie in *Operation of the Pamela high-level waste vitrification facility*, Nuclear Engineering and Design, 137 (1992), pp 147-151, state on page 149 that "One litre of HEWC containing ca. 100 g/liter of waste oxides, corresponding to an amount of ca. 3 kg of irradiated heavy metal, results in 0.2 liter of glass product after vitrification." HEWC stands for high enriched waste concentrate. Wiese and Demonie show in their Figure 2 that 106 grams of waste oxide reside in 0.2 liter of glass which has a mass of 0.49 kg; this yields a waste oxide loading of 22%. P.C. Upson in *Highly Active Liquid Waste Management at Sellafield*, Progress in Nuclear Energy, Vol. 13, No. 1, pp 31-47, shows in Table 1 that HLW oxides constitute 25% of the vitrified waste. These three references report HLW as oxides to glass.

Ferrous sulfamate is sometimes added to reduce Pu(VI) to Pu(III) by the ferrous ion, and is also widely used to reduce Pu(IV) to Pu(III). The sulfamate ion is known as a holding reductant and destroys nitrite ion that would otherwise catalytically oxidize Fe(II) to Fe(III) which is not desired. In a typical process 100 lb (45.45 kg) of iron sulfate is formed for each metric ton of uranium processed, or about 4.5% by mass based on uranium (J.T. Long, *Engineering for Nuclear Fuel Reprocessing*, 1978, American Nuclear Society, page 173). Ferrous sulfamate was used in the first Purex flow sheets at Hanford and Savannah River. The process was satisfactory in all respects except its addition of extraneous, nonvolatile components to the waste (M. Benedict, T.H. Pigford, H.W. Levi, *Nuclear Chemical Engineering*, 1981, page 487). Frequently plutonium is reduced to the trivalent state with ferrous sulfamate, in which the reducing and valence stabilizing properties of this compound are combined (J.M. Cleveland, *The Chemistry of Plutonium*, 1979, page 55). Given that the stoichiometry of the plutonium reduction is one-to-one, add ferrous sulfamate in excess by a factor of 2 based on the plutonium moles.

Given a mass of an isotope in a spent fuel assembly, multiply the mass by the oxide-mass-multiplier-for-element value in the tables here to obtain the mass of the isotope as oxide. Used molecular weight of oxygen as 16.0 even though it is reported as 15.9994 for the average isotopic composition of 99.759 atomic % for ^{16}O, 0.0374 atomic % for ^{17}O, and 0.2039 atomic % for ^{18}O; see Benedict, Pigford and Levi, 1981, *Nuclear Chemical Engineering*, 2-nd, Appendix C, page 942. Numbers below are truncated, not rounded. Half life data are from Benedict, Pigford and Levi except for Pu-241 which is from D.C. Kocher, *Radioactive Decay Data Tables: A Handbook of Decay Data for Application to Radiation Dosimetry and Radiological Assessments*, 1981, DOE/TIC-11026, page 211.

The tables that follow here, there are two, illustrate the calculation of the spent-fuel assembly composition as oxides which yields an average element-to-oxide mass multiplier, and document the individual element-to-oxide mass multipliers. The references for the isotope information then follows.

Table 1. Spread Sheet Illustrating the Calculation of Fission-Product Oxide Masses from Elemental Masses from a 60 GWd/ton Assembly.

File: HLW_oxide_mass_nGlass60.xls; August 16, 2010, page 1 of 2.

The "element multiplier to HLW" represents the separation process; 0.1% of uranium goes to HLW so it gets a 0.001, 0.1% of plutonium goes to HLW. Am-241 and Pu-241 are the same oxide mole weight so keep 50% of the Pu-241 because Pu-241 + Am-241 is constant mass to correspond to two Pu-241 half lives of 13.2 yrs; keeping 50% corresponds to 26.4-year old spent fuel being reprocessed.

Masses in grams from: **Spread Sheet for 60 GWd/MT at 4.5% initial U235; June 1, 2010**

Oxide multiplier reference: *Fission Product Oxides in Glass*, fission_product_oxides_08June2010.doc.

Assembly mass reference: CSNF_Composition_60GWd.xls; June 2, 2010.

Element	Isotope	In one assembly: mass, grams	Element multiplier to HLW, grams	HLW mass, grams	Oxide multiplier to HLW, grams	Oxide mass to glass, grams
Americium	Am-243	160.7	1.000	160.7	1.132	181.9
Barium	Ba-138	1008	1.000	1008.0	1.116	1124.9

Cadmium	Cd-110	42.8	1.000	42.8	1.146	49.0
Cadmium	Cd-111	20.0	1.000	20.0	1.144	22.9
Cadmium	Cd-112	10.0	1.000	10.0	1.143	11.4
Cadmium	Cd-114	10.8	1.000	10.8	1.140	12.3
Cadmium	Cd-116	3.8	1.000	3.8	1.138	4.3
Cerium	Ce-140	971.3	1.000	971.3	1.228	1192.8
Cerium	Ce-142	861.4	1.000	861.4	1.225	1055.2
Cerium	Ce-144	177.7	1.000	177.7	1.222	217.1
Cesium	Cs-133	810.9	1.000	810.9	1.060	859.6
Cesium	Cs-135	305.9	1.000	305.9	1.059	323.9
Cesium	Cs-137	933.8	1.000	933.8	1.058	988.0
Curium	Cm-245	4.5	1.000	4.5	1.131	5.1
Lanthanum	La-139	937.0	1.000	937.0	1.173	1099.1
Molybdenum	Mo-95	506.7	1.000	506.7	1.505	762.6
Molybdenum	Mo-97	619.9	1.000	619.9	1.495	926.8
Molybdenum	Mo-98	638.6	1.000	638.6	1.490	951.5
Molybdenum	Mo-100	729.7	1.000	729.7	1.480	1080.0
Neodymium	Nd-143	508.6	1.000	508.6	1.167	593.5
Neodymium	Nd-144	920.9	1.000	920.9	1.166	1073.8
Neodymium	Nd-145	480.2	1.000	480.2	1.165	559.4
Neodymium	Nd-146	576.8	1.000	576.8	1.164	671.4
Neptunium	Np-237	382.9	1.000	382.9	1.135	434.6
Palladium	Pd-105	324.6	1.000	324.6	1.000	324.6
Plutonium	Pu-238	204.6	0.001	0.2	1.133	0.2
Plutonium	Pu-239	2628.0	0.001	2.6	1.133	3.0
Plutonium	Pu-240	1345.0	0.001	1.3	1.133	1.5
Plutonium	Pu-241	833.3	0.500	416.7	1.132	471.6
Plutonium	Pu-242	506.4	0.001	0.5	1.132	0.6
Praseodymium	Pr-141	829.2	1.000	829.2	1.227	1017.4
Rhodium	Rh-103	291.9	1.000	291.9	1.000	291.9
Rubidium	Rb-85	72.9	1.000	72.9	1.000	72.9
Rubidium	Rb -87	181.1	1.000	181.1	1.000	181.1
Ruthenium	Ru-101	597.1	1.000	597.1	1.317	786.4
Ruthenium	Ru-102	647.7	1.000	647.7	1.314	851.1
Ruthenium	Ru-104	460.2	1.000	460.2	1.310	602.9
Selenium	Se-79	3.6	1.000	3.6	1.307	4.7
Silver	Ag-109	62.1	1.000	62.1	1.147	71.2
Strontium	Sr-90	385.5	1.000	385.5	1.178	454.1
Technetium	Tc-99	569.7	1.000	569.7	1.323	753.7
Tellurium	Te-130	293.4	1.000	293.4	1.246	365.6
Tin	Sn-116	2.6	1.000	2.6	1.276	3.3
Tin	Sn-117	3.4	1.000	3.4	1.274	4.3
Tin	Sn-118	2.9	1.000	2.9	1.271	3.7
Tin	Sn-119	3.0	1.000	3.0	1.269	3.8
Tin	Sn-122	3.9	1.000	3.9	1.263	4.9
Tin	Sn-124	6.4	1.000	6.4	1.258	8.1

Tin	Sn-126	15.0	1.000	15.0	1.254	18.8
Uranium	U-234	72.0	0.001	0.1	1.181	0.1
Uranium	U-235	2430.0	0.001	2.4	1.181	2.9
Uranium	U-236	2777.0	0.001	2.8	1.180	3.3
Uranium	U-238	391800	0.001	391.8	1.179	461.9
Yttrium	Y-89	321.5	1.000	321.5	1.269	408.0
Zirconium	Zr-91	421.1	1.000	421.1	1.352	569.3
Zirconium	Zr-92	482.3	1.000	482.3	1.348	650.1
Zirconium	Zr-93	526.0	1.000	526.0	1.344	706.9
Zirconium	Zr-94	586.3	1.000	586.3	1.340	785.6
Zirconium	Zr-96	614.2	1.000	614.2	1.333	818.7

HLW element mass sum =	20148.9	grams per one assembly.
HLW oxide mass sum =	24909.5	grams per one assembly.
Element-to-oxide mass multiplier =	1.236	
MoO3 oxide mass =	2494.9	grams per one assembly.
MoO3 oxide wt % =	10.0	in HLW oxide mass.
Noble metal (RuO2 +Pd+Rh+Te) mass sum =	3222.4	grams per one assembly.
Noble metal (RuO2+Pd+Rh+Te) element wt % =	12.9	in HLW oxide mass.
Assembly heavy metal mass, kilograms =	430.0	

Table 2. Isotope Information Illustrating the Calculation of Individual Oxide Mass Multipliers for Each Isotope.

Element Name	Common Isotope Name	Isotope mole wt.	Half life	Oxide Form	Oxide mole wt.	Oxide mass multiplier for element
Americium	Am-243	243.1	7950 yr	AmO_2	275.1	1.132
Barium	Ba-138	138.3	stable	BaO	154.3	1.116
Cadmium	Cd-110	109.9	stable	CdO	125.9	1.146
Cadmium	Cd-111	110.9	stable	CdO	126.9	1.144
Cadmium	Cd-112	111.9	stable	CdO	127.9	1.143
Cadmium	Cd-114	113.9	stable	CdO	129.9	1.140
Cadmium	Cd-116	115.9	stable	CdO	131.9	1.138
Cerium	Ce-140	139.9	stable	CeO_2	171.9	1.228
Cerium	Ce-142	141.9	stable	CeO_2	173.9	1.225
Cerium	Ce-144	143.9	**0.78 yr**	CeO_2	175.9	1.222
Cesium	Cs-137	136.9	30.0 yr	Cs_2O	289.8	1.058
Cesium	Cs-133	132.9	stable	Cs_2O	281.8	1.060
Cesium	Cs-135	134.9	3E6 yr	Cs_2O	285.8	1.059
Curium	Cm-245	245.1	9300 yr	CmO_2	277.1	1.131
Lanthanum	La-139	138.9	stable	La_2O_3	325.8	1.173
Molybdenum	Mo-95	94.9	stable	MoO_3	142.9	1.505
Molybdenum	Mo-97	96.9	stable	MoO_3	144.9	1.495
Molybdenum	Mo-98	97.9	stable	MoO_3	145.9	1.490
Molybdenum	Mo-100	99.9	stable	MoO_3	147.9	1.480
Neodymium	Nd-143	142.9	stable	Nd_2O_3	333.8	1.167
Neodymium	Nd-144	143.9	stable	Nd_2O_3	335.8	1.166
Neodymium	Nd-145	144.9	stable	Nd_2O_3	337.8	1.165
Neodymium	Nd-146	145.9	stable	Nd_2O_3	339.8	1.164
Neptunium	Np-237	237.0	2.14E6 yr	NpO_2	269.0	1.135
Palladium	Pd-105	104.9	stable	Pd	104.9	1.000
Plutonium	Pu-239	239.0	24400 yr	PuO_2	271.0	1.133
Plutonium	Pu-240	240.0	6580 yr	PuO_2	272.0	1.133
Plutonium	Pu-241	241.0	14.4 yr	PuO_2	273.0	1.132
Plutonium	Pu-242	242.0	3.8E5 yr	PuO_2	274.0	1.132
Praseodymium	Pr-141	140.9	stable	PrO_2	172.9	1.227
Rhodium	Rh-103	102.9	stable	Rh	102.9	1.000
Ruthenium	Ru-101	100.9	stable	RuO_2	132.9	1.317
Ruthenium	Ru-102	101.9	stable	RuO_2	133.9	1.314
Ruthenium	Ru-103	102.9	**0.108 yr**	RuO_2	134.9	1.310
Ruthenium	Ru-104	103.9	stable	RuO_2	135.9	1.307
Selenium	Se-79	78.9	6.4E4	SeO_2	110.9	1.406
Silver	Ag-109	108.9	stable	AgO	124.9	1.147
Strontium	Sr-90	89.9	28.1 yr	SrO	105.9	1.178
Technetium	Tc-99	99.0	2.12E yr	TcO_2	131.0	1.323

Element Name	Common Isotope Name	Isotope mole wt.	Half life	Oxide Form	Oxide mole wt.	Oxide mass multiplier for element
Tellurium	Te-130	129.9	stable	TeO_2	161.9	1.246
Tin	Sn-116	115.9	stable	SnO_2	147.9	1.276
Tin	Sn-117	116.9	stable	SnO_2	148.9	1.273
Tin	Sn-118	117.9	stable	SnO_2	149.9	1.271
Tin	Sn-119	118.9	stable	SnO_2	150.9	1.269
Tin	Sn-122	121.9	stable	SnO_2	153.9	1.262
Tin	Sn-124	123.9	stable	SnO_2	155.9	1.258
Tin	Sn-126	125.9	1E5	SnO_2	157.9	1.254
Uranium	U-234	234.0	2.5E5 yr	U_3O_8	830.0	1.182
Uranium	U-235	235.0	7.1E8 yr	U_3O_8	833.0	1.181
Uranium	U-236	236.0	2.39E7 yr	U_3O_8	836.0	1.181
Uranium	U-238	238.0	4.51E9 yr	U_3O_8	842.0	1.179
Yttrium	Y-89	88.9	stable	Y_2O_3	225.8	1.269
Zirconium	Zr-91	90.9	stable	ZrO_2	122.9	1.352
Zirconium	Zr-92	91.9	stable	ZrO_2	123.9	1.348
Zirconium	Zr-93	92.9	1.5E6 yr	ZrO_2	124.9	1.344
Zirconium	Zr-94	93.9	stable	ZrO_2	125.9	1.340
Zirconium	Zr-96	95.9	stable	ZrO_2	127.9	1.333

References for Element Information Used in Table 2:

Americium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 977; Oxide chosen as the dioxide because the electronic ground state of plutonium is $5f^6 7s^2$ which yields PuO_2 and the ground state of americium is $5f^7 7s^2$; since they both have an s^2 ground state use AmO_2. See J.M. Cleveland, The Chemistry of Plutonium, 1979, page 5.

Barium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 977; Oxide form from Table 9.4, Lange’s Handbook of Chemistry, 14-th, Table 3.2, page 3.19.

Cadmium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 955 - 956; Oxide form from Lange’s Handbook of Chemistry, 14-th, Table 3.2, page 3.22.

Cerium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 963; Oxide form from Lange’s Handbook of Chemistry, 14-th, Table 3.2, page 3.25.

Cesium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 962; Oxide form from Lange’s Handbook of Chemistry, 14-th, Table 3.2, page 3.26.

Curium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 962; Oxide form from Nuclear Chemical Engineering, 2-nd, Table 9.4, page 410.

Lanthanum: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 963; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.36.

Molybdenum: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 952; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.41.

Neodymium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 964; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.41.

Palladium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 954. Use as element.

Plutonium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 976; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.44.

Praseodymium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 964; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.47.

Rhodium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 953 – 954. Use as element.

Ruthenium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 953; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.49; also from AREVA's May 11, 2010 presentation, slide 13, noble metal mass used RuO_2, not the element.

Selenium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 948; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.49 – 3.50.

Strontium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 951; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.56.

Technetium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 953; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.57(?).

Tellurium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 953; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.57(?).

Tin: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 948; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.59.

Uranium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 975-976; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.60.

Yttrium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 951; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.62.

Zirconium: Isotope information from Nuclear Chemical Engineering, 2-nd, Appendix C, page 951-952; Oxide form from Lange's Handbook of Chemistry, 14-th, Table 3.2, page 3.63.

IV. Calculation Details of HLW Loading in Borosilicate Glass

The section describes the detailed calculation of HLW loading in glass from a burnup of 60 GWd/ton fuel that can be incorporated into borosilicate glass as a final waste form. The information required to perform this calculation comes from the open literature and does not require a description of the process chemicals added during reprocessing.

The AREVA presentation (by Paul Murray) of May 11, 2010, to the Nuclear Waste Technical Review Board Staff in Arlington gives a reference glass formulation that is quite similar to one published by F. Frizon, S. Gin, C. Jegou, *Mass Transfer Phenomena in Nuclear Waste Packages*, pages 39-46, in L.Q. Wang (Ed.): *Advances in Transport Phenomena*, 2009, Springer-Verlag. These two compositions are given here in the following table:

Table 1. Compositions and Waste Loadings for Borosilicate Glass; Two References.

Oxide in Glass	Nominal composition from Frizon, et al., Table 1, wt %	Reference composition from Murray, slide 13, wt %
SiO_2	45.1	45.2
B_2O_3	13.9	13.9
Al_2O_3	4.9	4.9
Na_2O	9.8	9.8
Li_2O	2.0	2.0
ZnO	2.5	2.5
CaO	4.0	4.0
Fe_2O_3	2.9	2.9
NiO	0.4	0.4
Cr_2O_3	0.5	0.5
Radioactive waste oxides as: FP + actinides + noble metals and Zr fines	12.8 + 0.9 = 13.7	13.7

Disposal glasses are obtained by melting the waste oxides with additives (probably called "frit") such as SiO_2, B_2O_3, Al_2O_3, Na_2O, P_2O_5 and CaO (from M. Benedict, T.H. Pigford, H.W. Levi, *Nuclear Chemical Engineering*, page 580, 1981). These additives listed here are **bolded** in column 1 (P_2O_5 is absent) and sum to 77.7%. The sum of the other non-radioactive waste constituents listed, which are lithium, zinc, iron, nickel and chromium, is 8.3%. These other constituents probably come from process chemicals added during reprocessing and end up as the indicated oxides in the final glass. Note that these columns sum very close to 100%; 99.7% and 99.8% respectively. Thus the *radioactive waste plus process chemicals must be 100.0% – 77.7% ≈ 22.3%*. Since the radioactive *waste oxides* total 13.7%, use this as a typical HLW (as fission products) loading and ignore the specifics of the process chemicals; this glass recipe implicitly takes into account process chemicals.

Conclusion: Use an HLW loading as fission-product oxides plus actinides that get through reprocessing, typically 0.1% loss, plus noble metals as elements, equal to 13.7 weight %.

There are composition limits in glass for molybdenum trioxide and the noble metals. How these limits enter into the calculation are described here. Specify that the final glass must contain MoO_3 in the range of 1 to 2 weight % and noble metals (RuO_2 + Pd + Rh + Te, ruthenium oxide + palladium + rhodium + tellurium) in the range of 1.0 to 1.5 weight %. (Y. Inagaki, T. Iwasaki, S. Sato, T. Ohe, K. Kato, S. Torikai, Y. Niibori, S. Nagasaki, K. Kitayama, *LWR High Burn-Up Operation and MOX Introduction; Fuel Cycle Performance from the Viewpoint of Waste Management*, Journal of Nuclear Science and Technology, Vol. 46, No. 7, p. 677-689, 2009, specifically see page 679). Use a 1.5 weight % MoO_3 concentration limit and a 1.25 weight % noble metal concentration limit in glass (as the single oxide and elements as noted above).

Consider that the glass waste canister is 2 feet in diameter by 10 feet long; the volume is 0.89 m^3 (890 liters of glass); this is used in NUWASTE. Use a final glass density of 2.75 MT/m^3 (from Frizon, et al., mean value from page 43). The final waste glass mass in one canister is 2.4475 MT, or 2447.5 kilograms.

The HLW oxide mass in this glass waste canister is 13.7% × 2447.5 = 335.3 kilograms. To obtain the HLW elemental mass divide by 1.236 to obtain 271.3 kilograms, or 0.271 MT elemental mass per canister (the 1.236 factor comes from *HLW_oxide_mass_nGlass60.xls;* August 16, 2010). Now check to see if MoO_3 and the noble metals compositions "fit" in this glass.

Given that there will be 335.3 kilograms HLW oxide in one canister and that the MoO_3 is 10.0 wt % of this HLW oxide mass (from *HLW_oxide_mass_nGlass60.xls*; August 16, 2010), the MoO_3 mass loading in the final glass is 10.0% × 335.3 = 33.5 kilograms; or the final MoO_3 loading in the glass is 33.5/2447.5 × 100 = 1.37 wt %. MoO_3 "fits" within this glass since this loading is less than the 1.50 wt% maximum.

Again, given that there will be 335.3 kilograms HLW oxide in one canister and that the noble metal (element) sum is 12.9 wt % of this HLW oxide mass (from *HLW_oxide_mass_nGlass60.xls*; August 16, 2010), the noble metal mass loading in the final glass is 12.9% × 335.3 = 43.3 kilograms; or the final noble metal loading in the glass is 43.3/2447.5 × 100 = 1.77 wt %. The noble metals ***do not "fit"*** within this glass because the limit is 1.25 weight %. In order for the noble metals to not exceed 1.25%, the HLW waste loading must be reduced by 1.25/1.77 = 0.706; this yields an HLW oxide mass loading of 0.706 × 0.335 = 0.2367 MT per canister, or an HLW elemental mass loading of 0.1915 MT per canister (this is the NUWASTE input parameter).

Given that one assembly produces 24.9 kilograms of HLW oxide (from *HLW_oxide_mass_nGlass60.xls*, August 16, 2010), a glass canister contains the HLW oxide mass from 236.7/24.9 = 9.5 (round to 10) assemblies (from the 0.2367 MT oxide loading immediately above). These 9.5 assemblies contain 4.1 MT initial heavy metal (430 kg metal per assembly, from *CSNF_Composition_60GWd.xls*, June 2, 2010). Thus the volume of glass produced per metric ton of initial heavy metal is 890 liters (in one canister)/4.1 = 217 liters.

The HLW elemental loading in one glass canister is 0.1915 MT per canister and is controlled by the noble metal loading. This HLW glass canister loading produces 217 liters of glass per metric ton initial heavy metal. If the noble-metal concentration limited in not invoked the HLW elemental loading in one glass canister is 0.271 MT.

References (Section IV):

F. Frizon, S. Gin, C. Jegou, *Mass Transfer Phenomena in Nuclear Waste Packages*, pages 39-46, in L.Q. Wang (Ed.): *Advances in Transport Phenomena*, 2009, Springer-Verlag, see pages 43 & 44.

Y. Inagaki, T. Iwasaki, S. Sato, T. Ohe, K. Kato, S. Torikai, Y. Niibori, S. Nagasaki, K. Kitayama, *LWR High Burn-Up Operation and MOX Introduction; Fuel Cycle Performance from the Viewpoint of Waste Management*, Journal of Nuclear Science and Technology, Vol. 46, No. 7, p. 677-689, 2009, see page 679.

M. Benedict, T.H. Pigford, H.W. Levi, *Nuclear Chemical Engineering*, page 580, 1981

B. Kirstein, *Fission Product Oxides in Glass*, File: fission_product_oxides_07June2010.doc. This document describes the expected fission product oxides and the element-to-oxide mass multiplier for each fission product, use the multipliers in the table.

B. Kirstein, *HLW_oxide_mass_nGlass60.xls*, June 3, 2010. This Excel spread sheet calculates a lumped element-to-oxide mass multiplier based on a specified fission-product composition.

G. Rowe, *CSNF_Composition_60GWd.xls*, June 2, 2010. This Excel spread sheet documents the fission-product spectrum for spent nuclear fuel burned to 60 GWd/ton. This spread sheet was slightly edited by Kirstein.

V. Code Listing for GLASSVOL.BAS

The following BASIC code should run from DOS on any BASIC interpreter, just load it and run. Then go to Word and retrieve the output file: GLASINFO.TXT and print it. This code is not a “production” code, it’s a once-run then print-it code.

```
100 REM *******
110 CODE$="GLASSVOL.BAS"
120 VER$="8-16-10 @ 1016"
130 REM Specify the glass density, gm/cc
140 RGLASS=2.75
150 REM Specify the HLW oxide loading as a weight % of the final glass.
160 HLWOXIDE=13.7
170 REM Specify the MoO3 loading in the HLW oxide, wt %.
```

```
180 MOO3OXID=10.1
190 REM Specify the MoO3 weight % in the final glass.
200 MOLY3MAX=1.5
210 REM Specify the noble metal loading (RuO2 + Rh + Pd) weight % in the
220 REM final glass.
230 NOBLES=10.3
240 REM Specify the maximum noble metal loading in the final glass mass,
250 REM wt %.
260 NOBLEMAX=1.25
270 REM Specify the glass canister size in cubic meters.
280 CANVOL=.89
290 REM Specifiy the element-to-oxide mass multiplier.
300 ELOXIDE=1.207
310 REM Specify the fuel assembly heavy metal mass, kilograms.
320 ASSYMASS=430!
330 REM Specify the fission-product oxide mass per assembly, kilograms.
340 FPMASS=24.4
350 REM Set up an identification string.
360 ID$="You enter some identification."
370 REM Initialize some parameters.
380 HLW1=1!
390 HLW2=1!
400 JUST=1
410 REM Set a screen tab.
420 NT=58
430 REM Set some output tabs.
440 NT1=55
450 NT2=55
460 NT3=60
470 CLS
480 PRINT"Input edit screen for ";CODE$;
490 PRINT", version of ";VER$
500 PRINT:PRINT"1.  Glass density, gm/cc = ";
510 PRINT TAB(NT);
520 PRINT USING"##.##";RGLASS
530 PRINT"2.  HLW oxide loading in glass, wt % = ";
540 PRINT TAB(NT);
550 PRINT USING"##.##";HLWOXIDE
560 PRINT
570 PRINT"3.  MoO3 loading in HLW oxide mass, wt % = ";
580 PRINT TAB(NT);
590 PRINT USING"##.##";MOO3OXID
600 PRINT"4.  Maximum MoO3 loading in final glass, wt % = ";
610 PRINT TAB(NT);
620 PRINT USING"##.##";MOLY3MAX
630 PRINT
```

```
640 PRINT"5.  Noble metal loading in HLW oxide mass, wt % = ";
650 PRINT TAB(NT);
660 PRINT USING"##.##";NOBLES
670 PRINT"6.  Maximum noble metal loading in final glass, wt % = ";
680 PRINT TAB(NT);
690 PRINT USING"##.##";NOBLEMAX
700 PRINT
710 PRINT"7.  Glass canister volume, m^3 = ";
720 PRINT TAB(NT);
730 PRINT USING"##.##";CANVOL
740 PRINT"8.  Element-to-oxide mass multiplier = ";
750 PRINT TAB(NT);
760 PRINT USING"##.###";ELOXIDE
770 PRINT"9.  Assembly heavy metal mass, kilograms = ";
780 PRINT TAB(NT);
790 PRINT USING"####";ASSYMASS
800 PRINT"10. Fission product oxide mass per assembly, kg = ";
810 PRINT TAB(NT);
820 PRINT USING"###.#";FPMASS
830 PRINT
840 PRINT"11. The ID string is: ";
850 PRINT ID$
860 PRINT:INPUT"Want to change any";AN$
870 TEST$=LEFT$(AN$,1)
880 IF TEST$="n" OR TEST$="N" GOTO 1160
890 PRINT:INPUT"Enter the item number to be changed";ILINE
900 ON ILINE GOTO 910,930,950,970,990,1020,1050,1070,1090,1110,1130
910 PRINT:INPUT"Enter the glass density, gm/cc";RGLASS
920 GOTO 470
930 PRINT:INPUT"Enter the HLW oxide loading in glass, wt % ";HLWOXIDE
940 GOTO 470
950 PRINT:INPUT"Enter the MoO3 loading in HLW oxide mass, wt %";MOO3OXID
960 GOTO 470
970 PRINT:INPUT"Enter the maximum MoO3 in final glass, wt %";MOLY3MAX
980 GOTO 470
990 PRINT:PRINT"Enter the noble metal loading in the HLW oxide";
1000 INPUT" mass, wt %";NOBLES
1010 GOTO 470
1020 PRINT:PRINT"Enter the maximum noble metal loading in the ";
1030 INPUT"final glass mass, wt %";NOBLEMAX
1040 GOTO 470
1050 PRINT:INPUT"Enter the glass canister volume, m^3";CANVOL
1060 GOTO 470
1070 PRINT:INPUT"Enter the element-to-oxide mass multiplier";ELOXIDE
1080 GOTO 470
1090 PRINT:INPUT"Enter the assembly mass, kilograms";ASSYMASS
```

```
1100 GOTO 470
1110 PRINT:INPUT"Enter the fission product oxide mass per assembly, kg";FPMASS
1120 GOTO 470
1130 PRINT"Enter an identification string:"
1140 INPUT ID$
1150 GOTO 470
1160 REM Save the initial HLW oxide loading.
1170 HLWSAVE=HLWOXIDE
1180 REM Calculate the glass mass in the canister, kilograms.
1190 GLASMASS=CANVOL*RGLASS*1000!
1200 REM Calculate the HLW oxide mass in this glass mass, kilograms.
1210 HLWOMASS=HLWOXIDE*GLASMASS/100!
1220 REM Calculate the HLW element mass in this glass mass, kilograms.
1230 HLWEMASS=HLWOMASS/ELOXIDE
1240 REM
1250 REM Calculate the MoO3 mass in the HLW oxide mass.
1260 MOO3MASS=MOO3OXID*HLWOMASS/100!
1270 REM Calculate the MoO3 wt % in the glass mass.
1280 MOO3WTPC=MOO3MASS*100!/GLASMASS
1290 REM
1300 REM Check to see if the MoO3 fits in the glass.
1310 IF MOO3WTPC<MOLY3MAX GOTO 1370
1320 REM The MoO3 does not fit.  Calculate the fraction reduction in HLW
1330 REM oxide mass put into the glass due to the MoO3 maximum.
1340 HLW1=MOLY3MAX/MOO3WTPC
1350 HLW1=.9999*HLW1
1360 REM
1370 REM Calculate the noble metal mass in the HLW oxide mass, kilograms.
1380 NOBLMASS=NOBLES*HLWOMASS/100!
1390 REM Calculate the nobel metal mass fraction in the glass mass.
1400 NOBLWTPC=NOBLMASS*100!/GLASMASS
1410 REM
1420 REM Check to see if the noble metals fit in the glass
1430 IF NOBLWTPC<NOBLEMAX GOTO 1490
1440 REM The noble metals do not fit.  Calculate the fraction reduction in
1450 REM HLW oxide mass put into the glass due to the noble metal maximum.
1460 HLW2=NOBLEMAX/NOBLWTPC
1470 HLW2=.9999*HLW2
1480 REM
1490 REM Check to see if a change in HLW oxide loading in glass has to
1500 REM made.  Check both MoO3 and noble metals, make the adjustment
1510 REM to the HLWOMASS to accommodate the one most of specification.
1520 REM
1530 REM Set a switch to keep track of who is responsible for the
1540 REM adjustment of the HLW oxide mass in the glass.
1550 REM
```

```
1560 REM If HLW1=1 and HLW2=1 then no change has to be made.
1570 REM
1580 IF HLW1=1! AND HLW2=1! GOTO 1810
1590 REM Fall through to here means a change has to be made.
1600 REM
1610 REM Check to see which is smaller.
1620 IF HLW2<HLW1 GOTO 1690
1630 REM Fall through to here means MoO3 is limiting.  Adjust the
1640 REM HLW oxide mass accordingly.
1650 HLWOMASS=HLW1*HLWOWMASS
1660 ADJUST=HLW1
1670 JUST=2
1680 GOTO 1720
1690 REM Branch to here means the noble metals are limiting.
1700 REM Adjust the HLW oxide mass accordingly.
1710 HLWOMASS=HLW2*HLWOMASS
1720 HLWOXIDE=HLWOMASS/GLASMASS*100!
1730 ADJUST=HLW2
1740 JUST=3
1750 REM Go recalculate everything.
1760 REM
1770 REM Reset HLW1 and HLW2.
1780 HLW1=1!
1790 HLW2=1!
1800 GOTO 1180
1810 REM Branch to here means everything fits in the glass, so
1820 REM calculate the glass volume per metric ton initial heavy
1830 REM metal.
1840 ASSYNUM=HLWOMASS/FPMASS
1850 MTIHM=ASSYNUM*ASSYMASS/1000!
1860 GLITERS=1000!*CANVOL/MTIHM
1870 REM
1880 REM Done.
1890 CLS
1900 PRINT"Output screen for ";CODE$;
1910 PRINT", version of ";VER$
1920 PRINT:PRINT"Glass mass in one canister, kilograms = ";
1930 PRINT TAB(NT1);
1940 PRINT USING"####.#";GLASMASS
1950 PRINT"HLW oxide mass in this glass mass, kilograms = ";
1960 PRINT TAB(NT1);
1970 PRINT USING"####.#";HLWOMASS
1980 PRINT"HLW element mass in this glass mass, kilograms = ";
1990 PRINT TAB(NT1);
2000 PRINT USING"####.#";HLWEMASS;
2010 PRINT", NUWASTE input."
```

```
2020 PRINT"HLW oxide loading in glass used in this calc, wt% =  ";
2030 PRINT TAB(NT1);
2040 PRINT USING"####.#";HLWOXIDE
2050 PRINT
2060 PRINT"MoO3 mass in the HLW oxide mass, kilograms = ";
2070 PRINT TAB(NT1);
2080 PRINT USING"####.#";MOO3MASS
2090 PRINT"MoO3 concentration in the glass, wt % = ";
2100 PRINT TAB(NT1);
2110 PRINT USING"###.##";MOO3WTPC
2120 PRINT
2130 PRINT"Noble metal mass in the HLW oxide mass, kilograms = ";
2140 PRINT TAB(NT1);
2150 PRINT USING"####.#";NOBLMASS
2160 PRINT"Noble metal concentration in the glass, wt % = ";
2170 PRINT TAB(NT1);
2180 PRINT USING"###.##";NOBLWTPC
2190 PRINT
2200 PRINT"# of assemblies that yield the fission product mass";
2210 PRINT TAB(NT2+2);
2220 PRINT USING"####";ASSYNUM;
2230 PRINT", in 1 canister."
2240 PRINT"Metric tons initial HM in these assemblies = ";
2250 PRINT TAB(NT2);
2260 PRINT USING"####.#";MTIHM
2270 PRINT"Glass volume per metric ton initial HM, liters = ";
2280 PRINT TAB(NT2);
2290 PRINT USING"####.#";GLITERS
2300 PRINT
2310 ON JUST GOTO 2320,2340,2370
2320 PRINT"No HLW oxide mass in glass adjustment made."
2330 GOTO 2440
2340 PRINT"HLW oxide mass adjusted to accommodate the MoO3 glass";
2350 PRINT" concentration."
2360 GOTO 2390
2370 PRINT"HLW oxide mass adjusted to accommodate the noble metal";
2380 PRINT" glass concentration."
2390 PRINT"The adjustment factor is ";
2400 PRINT USING"#.###";ADJUST;
2410 PRINT" of the intial input HLW oxide mass."
2420 PRINT"The initial HLW oxide loading in glass, wt %, was ";
2430 PRINT USING"###.#";HLWSAVE
2440 PRINT
2450 OPEN "O",#1,"GLASINFO.TXT"
2460 PRINT #1,"Output from ";CODE$;
2470 PRINT #1,", version of ";VER$
```

```
2480 PRINT #1,
2490 PRINT #1,"Printed from file: GLASINFO.TXT on ";DATE$;
2500 PRINT #1," at ";TIME$
2510 PRINT #1,
2520 PRINT #1,"Input variables for ID: ";ID$
2530 PRINT #1,
2540 PRINT #1,"1.  Glass density, gm/cc = ";
2550 PRINT #1,TAB(NT);
2560 PRINT #1,USING"##.##";RGLASS
2570 PRINT #1,"2.  HLW oxide loading in glass, wt % = ";
2580 PRINT #1,TAB(NT);
2590 PRINT #1,USING"##.##";HLWOXIDE
2600 PRINT #1,
2610 PRINT #1,"3.  MoO3 loading in HLW oxide mass, wt % = ";
2620 PRINT #1,TAB(NT);
2630 PRINT #1,USING"##.##";MOO3OXID
2640 PRINT #1,"4.  Maximum MoO3 loading in final glass, wt % = ";
2650 PRINT #1,TAB(NT);
2660 PRINT #1,USING"##.##";MOLY3MAX
2670 PRINT #1,
2680 PRINT #1,"5.  Noble metal loading in HLW oxide mass, wt % = ";
2690 PRINT #1,TAB(NT);
2700 PRINT #1,USING"##.##";NOBLES
2710 PRINT #1,"6.  Maximum noble metal loading in final glass, wt % = ";
2720 PRINT #1,TAB(NT);
2730 PRINT #1,USING"##.##";NOBLEMAX
2740 PRINT #1,
2750 PRINT #1,"7.  Glass canister volume, m^3 = ";
2760 PRINT #1,TAB(NT);
2770 PRINT #1,USING"##.##";CANVOL
2780 PRINT #1,"8.  Element-to-oxide mass multiplier = ";
2790 PRINT #1,TAB(NT);
2800 PRINT #1,USING"##.###";ELOXIDE
2810 PRINT #1,"9.  Assembly mass, kilograms = ";
2820 PRINT #1,TAB(NT);
2830 PRINT #1,USING"####";ASSYMASS
2840 PRINT #1,"10. Fission product oxide mass per assembly, kg = ";
2850 PRINT #1,TAB(NT);
2860 PRINT #1,USING"###.#";FPMASS
2870 PRINT #1,
2880 REM
2890 PRINT #1,"Calculated results:"
2900 PRINT #1,
2910 PRINT #1,"Glass mass in one canister, kilograms = ";
2920 PRINT #1,TAB(NT1);
2930 PRINT #1,USING"####.#";GLASMASS
```

```
2940 PRINT #1,"HLW oxide mass in this glass mass, kilograms = ";
2950 PRINT #1,TAB(NT1);
2960 PRINT #1,USING"####.#";HLWOMASS
2970 PRINT #1,"HLW element mass in this glass mass, kilograms = ";
2980 PRINT #1,TAB(NT1);
2990 PRINT #1,USING"####.#";HLWEMASS;
3000 PRINT #1,", NUWASTE input."
3010 PRINT #1,"HLW oxide loading in glass used in this calc, wt% =  ";
3020 PRINT #1,TAB(NT1);
3030 PRINT #1,USING"####.#";HLWOXIDE
3040 PRINT #1,
3050 PRINT #1,"MoO3 mass in the HLW oxide mass, kilograms = ";
3060 PRINT #1,TAB(NT1);
3070 PRINT #1,USING"####.#";MOO3MASS
3080 PRINT #1,"MoO3 concentration in the glass, wt % = ";
3090 PRINT #1,TAB(NT2);
3100 PRINT #1,USING"###.##";MOO3WTPC
3110 PRINT #1,
3120 PRINT #1,"Noble metal mass in the HLW oxide mass, kilograms = ";
3130 PRINT #1,TAB(NT1);
3140 PRINT #1,USING"####.#";NOBLMASS
3150 PRINT #1,"Noble metal concentration in the glass, wt % = ";
3160 PRINT #1,TAB(NT1);
3170 PRINT #1,USING"###.##";NOBLWTPC
3180 PRINT #1,
3190 PRINT #1,"# of assemblies that yield the fission product mass";
3200 PRINT #1,TAB(NT2);
3210 PRINT #1,USING"######";ASSYNUM;
3220 PRINT #1,", in 1 canister."
3230 PRINT #1,"Metric tons initial HM metal in these assemblies = ";
3240 PRINT #1,TAB(NT2);
3250 PRINT #1,USING"####.#";MTIHM
3260 PRINT #1,"Glass volume per metric ton initial HM, liters = ";
3270 PRINT #1,TAB(NT2);
3280 PRINT #1,USING"####.#";GLITERS
3290 PRINT #1,
3300 ON JUST GOTO 3310,3330,3360
3310 PRINT #1,"No HLW oxide mass in glass adjustment made."
3320 GOTO 3430
3330 PRINT #1,"HLW oxide mass adjusted to accommodate the MoO3 glass";
3340 PRINT #1," concentration."
3350 GOTO 3380
3360 PRINT #1,"HLW oxide mass adjusted to accommodate the noble metal";
3370 PRINT #1," glass concentration."
3380 PRINT #1,"The adjustment factor is ";
3390 PRINT #1,USING"#.###";ADJUST;
```

```
3400 PRINT #1," of the initial input HLW oxide mass."
3410 PRINT #1,"The initial HLW oxide loading in glass, wt %, was ";
3420 PRINT #1,USING"####.#";HLWSAVE
3430 PRINT #1,
3440 PRINT #1,"Refences * * * * * * * * * * * * * * * * * * * * * * * *";
3450 PRINT #1," * * * * *"
3460 PRINT #1,
3470 PRINT #1,"Initial HLW oxide loading in glass reference:"
3480 PRINT #1,"F. Frizon, S. Gin, C. Jegou, Mass Transfer Phenomena in";
3490 PRINT #1," Nuclear Waste Packages, "
3500 PRINT #1,"pages 39-64, in L.W. Wang (Ed.): Advances in Transport";
3510 PRINT #1," Phenomena, 2009,"
3520 PRINT #1,"Springer-Verlag."
3530 PRINT #1,
3540 PRINT #1,"Molybdenum and noble-metal maximum glass concentrations ";
3550 PRINT #1,"reference:"
3560 PRINT #1,"Y. Inagaki, T. Iwasaki, S. Sato, T. Ohe, K. Kato, S. Torikai,";
3570 PRINT #1," Y. Niibori, "
3580 PRINT #1,"S. Nagasaki, K. Kitayama, LWR High Burn-up Operation and ";
3590 PRINT #1,"MOX Introduction;"
3600 PRINT #1,"Fuel Cycle Performance from the Viewpoint of Waste Management";
3610 PRINT #1,", Journal"
3620 PRINT #1,"of Nuclear Science and Technology, Vol. 46, No. 7, p. 677-";
3630 PRINT #1,"689, 2009."
3640 PRINT #1,
3641 PRINT #1,"Pick up the output in GLASINFO.TXT."
3642 PRINT #1,
3650 END
```